





PARAB

Este programa calcula la orientación (respecto al norte magnético), la elevación y el desplazamiento de la polarización de antenas parabólicas en cualquier hemisferio, así como la distancia antena-satélite.

Definiciones:

- **Acimut:** ángulo que hay que girar la antena con respecto a la dirección del norte (o sur) geográfico.
- **Elevación:** ángulo que hay que elevar la antena con respecto al plano del suelo.
- **Desplazamiento de la polarización:** ángulo que hay que girar el dipolo con respecto al eje vertical para recibir correctamente la señal del satélite.
- **Declinación magnética:** diferencia entre la dirección del norte geográfico y la del norte magnético. Varía según la zona y cambia con el tiempo. En Internet se pueden encontrar calculadoras de la declinación magnética.
- **Orientación:** ángulo que hay que girar la antena con respecto al norte (o sur) indicado por la brújula.
- **Distancia al satélite:** los satélites de comunicaciones se hallan en una órbita situada en el plano del ecuador, con un radio de aproximadamente 42000 Km, 36000 sobre la superficie de la tierra, llamada órbita geoestacionaria o cinturón de Klarke. La distancia entre el satélite y la antena varía según dónde se encuentren ubicados éstos.



Desplazamiento de la polarización.

PROGRAMA

```
«
CLEAR DEG
"Latitud" " " INPUT OBJ→ 'LA' STO
LA 0 < IF THEN
"OPCION"{{"NORTE"1}{"SUR"2}}1
CHOOSE DROP 'NS' STO
END
LA 0 == IF THEN
0.001 'LA' STO
2 'NS' STO
END
"Longitud" " " INPUT OBJ→ 'LO' STO
LO 0 < IF THEN
"OPCION"{{"ESTE"1}{"OESTE" -1}}1
CHOOSE DROP 'LO' STO*
END
"Posición sat." " " INPUT OBJ→ 'LS' STO
LS 0 < IF THEN
"OPCION"{{"ESTE"1}{"OESTE" -1}}1
CHOOSE DROP 'LS' STO*
END
"Declinación mag." "" INPUT OBJ→ 'DM' STO
DM 0 < IF THEN
"OPCION"{{"ESTE" -1}{"OESTE"1}}1
CHOOSE DROP 'DM' STO*
NS 2 == IF THEN
-1 'DM' STO*
END
END
"Angulo offset" " " INPUT OBJ→ 'AO' STO
« LO LS - » 'FI' STO
NS 1 == IF THEN
« FI TAN LA SIN / ATAN 180 DM + + » 'ORI' STO
« LA COS FI COS * ACOS » 'b' STO
« b COS .151269 - b SIN / ATAN AO - » 'EL' STO
« FI NEG SIN LA AO + TAN / ATAN » 'PO' STO
« 1. b COS - .42 * 1 + √ 35786 * » 'DIS' STO
END
NS 2 == IF THEN
« LO LS - » 'FI' STO
« FI TAN LA SIN / ATAN NEG DM + » 'ORI' STO
« LA COS FI COS * ACOS » 'b' STO
« b COS .151269 - b SIN / ATAN AO - » 'EL' STO
« FI NEG SIN LA AO + TAN / ATAN NEG » 'PO' STO
« 1. b COS - .42 * 1 + √ 35786 * » 'DIS' STO
END
ORI 1. RND 1_° * "Orient" →TAG
EL 1. RND 1_° * "Elev" →TAG
PO 1. RND 1_° * "Pol" →TAG
DIS 3. RND 1_km * "Dist" →TAG
{ LO LA PO EL ORI FI AO DM LS b DIS NS } PURGE
»
PARAB STO
```

Entrada de datos

Cálculos hemisferio Norte

Cálculos hemisferio Sur

Presentación de resultados

DEG XYZ HEX R~ 'N' PRG
CHOME3 14:42 11:APR
Latitud
OPCION
NORTE
SUR
39.4
[CANCL] [OK]

DEG XYZ HEX R~ 'N' PRG
CHOME3 14:43 11:APR
Longitud
OPCION
ESTE
OESTE
0.54
[CANCL] [OK]

DEG XYZ HEX R~ 'N' PRG
CHOME3 14:44 11:APR
Posición sat.
OPCION
ESTE
OESTE
19.24
[CANCL] [OK]

DEG XYZ HEX R~ 'N' PRG
CHOME3 14:45 11:APR
Declinación mag.
OPCION
ESTE
OESTE
1.24
[CANCL] [OK]

DEG XYZ HEX R~ 'N' PRG
CHOME3 14:45 11:APR
Angulo offset
26
[ON] [LS] [LO] [OS] [LA] [GMS]

DEG XYZ HEX R~ 'N' PRG
CHOME3 14:46 11:APR
7:
6:
5:
4: Orient:(151.6_8)
3: Elev:(14.4_8)
2: Pol:(8.9_8)
1: Dist:(37748.765_km)
[GMS] [PARAB] [ALTER] [FRIIS] [CASDI]